

# Stata 中的空间自回归模型

刘迪 @StataCorp

# 目录

- 1 为什么需要空间自回归模型？
- 2 实例：得克萨斯州犯罪率
- 3 Stata 之 **Sp** 系列命令的概述

# 目录

- 1 为什么需要空间自回归模型？
- 2 实例：得克萨斯州犯罪率
- 3 Stata 之 **Sp** 系列命令的概述

## 为什么需要空间自回归模型？

*Everything is related to everything else, but near things are more related than distant things. — Tobler*

- 圣安东尼奥的犯罪率可能影响休斯顿的犯罪率。然而，巴黎的犯罪率却很难受到它的影响，因为这两个城市距离得太遥远了。
- 纽约市某个街区的白血病发病率的上升可能会影响周边街区的白血病发生率。虽然白血病没有传染性，但是一些无法观测到的因素可能是空间相关的（例如环境污染，核辐射）。

## 为什么需要空间自回归模型？

- 经典的线形回归模型无法将空间的相关性考虑进去。
- 空间自回归模型（Spatial Autoregression, SAR）可以包括空间相关的因变量和空间相关的误差项。
- SAR 模型帮助我们理解空间距离的影响。这个空间距离既可以是地理意义上的，也可以是抽象的（例如社交网络）。


SAR 模型帮助我们理解如下问题：

某个空间单位的自变量  $X$  的变化如何影响其它空间单位的因变量  $y$  ？

# 目录

- 1 为什么需要空间自回归模型？
- 2 实例：得克萨斯州犯罪率
- 3 Stata 之 **Sp** 系列命令的概述

# 实例：得克萨斯州犯罪率



- 高犯罪率县的周边大部分是高犯罪率地区。
- 低犯罪率县的周边大部分是低犯罪率地区。

## 实例：得克萨斯州犯罪率

- 一些学者试图分析德州的失业率对犯罪率的影响。他们假设一个县的犯罪率会影响其它的县。
- 他们需要
  - ① 建立一个模型，其中一个县的犯罪率可以影响其它县的犯罪率。
  - ② 估算出一个县失业率的变化如何影响其它县的犯罪率。

# 空间自回归模型

- 我们需要一个模型以便考虑到犯罪率的空间相关性。

$$\text{hrate}_i = \beta_0 + \beta_1 * \text{unemployment}_i + \lambda * \text{hrate}_{\text{neighbors of } i} + \text{errors}$$

- 我们用一个矩阵  $W$  来定义"neighbors".  $W$  即县与县之间的空间相关性的强弱。

$$\text{hrate} = \beta_0 + \beta_1 * \text{unemployment} + \lambda * W * \text{hrate} + \text{errors}$$

## 利用空间权重矩阵定义空间关系的强弱


一个空间权重矩阵  $W$  总结了  $n$  个空间单位之间的关系强弱。

$$W = \begin{bmatrix} 0 & w_{12} & w_{13} & \dots & w_{1n} \\ w_{21} & 0 & w_{23} & \dots & w_{2n} \\ \vdots & & \ddots & & \vdots \\ \vdots & & & \ddots & \vdots \\ w_{n1} & w_{n2} & & \dots & 0 \end{bmatrix}_{n \times n}$$

- $W$  是一个  $n \times n$  的矩阵且  $w_{ij} \geq 0$ 。
- $w_{ij}$  反映了空间单位  $i$  对空间单位  $j$  的影响。
- 我们不考虑自身的影响，所以对角线元素  $w_{ii}$  都为零。

## 空间邻接矩阵 (Spatial contiguity matrix)

- 我们先看三个紧邻彼此的县。简单起见，我们叫它们 A, B, 和 C。




- 一种空间权重矩阵叫做邻接矩阵，其中接壤的两个县被定义为邻居。例如，

$$W = \begin{array}{c} \\ A \\ B \\ C \end{array} \begin{array}{ccc} A & B & C \\ 0 & 1 & 0 \\ 1 & 0 & 1 \\ 0 & 1 & 0 \end{array}$$

## 空间相反距离矩阵 (Spatial inverse-distance matrix)

- 在这个例子中，我们假设距离是没有方向之分的（A 到 B 的距离和 B 到 A 的距离是一样的）。



- 我们可以基于距离的倒数来建立空间相反距离矩阵。

$$W = \begin{array}{c|ccc} & A & B & C \\ \hline A & 0 & 1/50 & 1/125 \\ B & 1/50 & 0 & 1/75 \\ C & 1/125 & 1/75 & 0 \end{array}$$

## 使用 `spmatrix` 来定义空间权重矩阵

- 让我们先看看数据，其中 `_CX` 和 `_CY` 是一个县地理中心的经度和纬度。

```
. use texas, clear  
(S.Messner et al.(2000), U.S southern county homicide rates in 1990)  
. list _CX _CY cname hrate unemployment in 1/7, abbreviate(12)
```

	<code>_CX</code>	<code>_CY</code>	<code>cname</code>	<code>hrate</code>	<code>unemployment</code>
1.	-100.27156	36.275086	Lipscomb	0.00	1.73
2.	-101.8931	36.273254	Sherman	0.00	3.34
3.	-102.59591	36.27355	Dallam	18.31	2.28
4.	-101.35351	36.272304	Hansford	0.00	4.01
5.	-100.81561	36.273178	Ochiltree	3.65	4.87
6.	-100.81482	35.840515	Roberts	0.00	3.24
7.	-100.26948	35.839961	Hemphill	0.00	4.14

- 邻接矩阵定义两个接壤的县为邻居

```
. spmatrix create contiguity W
```

## 使用 `spregress` 来计算 SAR 模型

$$\text{hrate} = \beta_0 + \beta_1 * \text{unemployment} + \lambda * \mathbf{W} * \text{hrate} + \text{errors}$$

```
spregress hrate unemployment, dvarlag(W) gs2sls
```

- 选项 **dvarlag(W)** 意味着因变量 (dependent variable, **hrate**) 的空间权重平均值 (空间滞后项, spatial lag)。
- 选项 **gs2sls** 意味着估计方法为广义空间二阶最小二乘法 (generalized spatial two-stage least-squares estimator)。

# 使用 spregress 来计算 SAR 模型

```
. spregress hrate unemployment, dvarlag(W) gs2sls
(254 observations)
(254 observations (places) used)
(weighting matrix defines 254 places)
```

```
Spatial autoregressive model      Number of obs      =      254
GS2SLS estimates                  Wald chi2(2)       =     14.23
                                   Prob > chi2         =     0.0008
                                   Pseudo R2            =     0.0424
```

	hrate	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
hrate							
unemployment		.4584241	.152503	3.01	0.003	.1595237	.7573245
_cons		2.720913	1.653105	1.65	0.100	-.5191143	5.960939
W							
hrate		.3414964	.1914865	1.78	0.075	-.0338103	.7168031

```
Wald test of spatial terms:          chi2(1) = 3.18          Prob > chi2 = 0.0745
```

- 为了解释计算结果，我们需要首先理解**空间溢出效应**（**spatial spillover**）。

## 理解空间溢出效应 (I)

我们先做一个实验。请考虑以下一个具体的问题。


如果达拉斯的失业率增加到 20%，其它县的犯罪率会如何变化？

- 基于 **spregress** 的计算结果，我们可以用以下三步回答这个问题。
  - ① 基于原始的数据，预测各个县的犯罪率。
  - ② 改变达拉斯的失业率至 20%，再次预测各个县的犯罪率。
  - ③ 计算两次预测值的差值，将差值绘至得克萨斯州的地图上。

## 理解空间溢出效应 (II)

```
. preserve /* save data temporarily */  
.   
. /* Step 1: predict homicide rate using original data */  
. predict y0  
(option rform assumed; reduced-form mean)  
.   
. /* Step 2: change Dallas unemployment rate to 20%, and predict again*/  
. replace unemployment = 20 if cname == "Dallas"  
(1 real change made)  
. predict y1  
(option rform assumed; reduced-form mean)  
.   
. /* Step 3: Compute the prediction difference and map it*/  
. generate double y_diff = y1 - y0  
.   
. grmap y_diff, title("Global spillover")  
.   
. restore /* return to original data */
```

## 理解空间溢出效应 (III)



- 因为空间溢出效应不仅仅局限在和达拉斯直接接壤的邻居县, 我们将这种溢出效应称为全局溢出 (global spillover)。

## 空间溢出效应的秘密

$$\text{hrate} = \beta_0 + \beta_1 * \text{unemployment} + \lambda * W * \text{hrate} + \text{errors}$$

- 1 达拉斯的失业率上升至 20%。
- 2 达拉斯的犯罪率会随之升高。
- 3 和达拉斯接壤的邻居县随之受到影响，它们的犯罪率随之上升。
- 4 而与达拉斯邻居县接壤的邻居地区犯罪率也随之变化。这个过程不断扩散，直到所有的县都受到影响，但是距离达拉斯遥远的县收到的波及很微弱。

# 空间溢出效应的动态过程

## 使用 `estat impact` 来计算空间溢出效应

```
. estat impact
progress   :100%
Average impacts                Number of obs   =           254
```

	Delta-Method		z	P> z	[95% Conf. Interval]	
	dy/dx	Std. Err.				
direct unemployment	.4666538	.1539861	3.03	0.002	.1648466	.7684609
indirect unemployment	.1910068	.1565581	1.22	0.222	-.1158414	.497855
total unemployment	.6576605	.2519366	2.61	0.009	.1638739	1.151447

- $\frac{dy}{dx}$  是失业率对犯罪率的边际效应。
- 直接效应 (direct impact) 为 0.47, 所以一个县的失业率对本县的犯罪率的平均边际效应为 0.47。
- 间接效应 (indirect impact) 为 0.19, 所以一个县的失业率对其它县的犯罪率的平均边际效应为 0.19。

## 平均直接效应 (Average direct impact)

- 失业率的直接效应是指一个县的失业率对同一个县的犯罪率的边际效应。
- 达拉斯失业率的直接效应 (Direct impact, DI)

$$DI_{\text{Dallas}} = \frac{d \text{ hrate}_{\text{Dallas}}}{d \text{ unemployment}_{\text{Dallas}}}$$

- 平均直接效应是指所有县失业率的直接效应的平均值。

$$\text{Average direct impact}_{\text{Texas}} = \frac{1}{254} (DI_{\text{Dallas}} + DI_{\text{Houston}} + \dots)$$

## 平均间接效应 (Average indirect impact)

- 间接效应是指一个县的所有邻居县的失业率对于本县的犯罪率的边界效应的总和。
- 达拉斯的间接效应 (Indirect impact, IDI)

$$IDI_{\text{Dallas}} = \frac{d \text{ hrate}_{\text{Dallas}}}{d \text{ unemployment}_{\text{Houston}}} + \frac{d \text{ hrate}_{\text{Dallas}}}{d \text{ unemployment}_{\text{Brazos}}} + \dots$$

(where the "... **does not include**  $\frac{d \text{ hrate}_{\text{Dallas}}}{d \text{ unemployment}_{\text{Dallas}}}$ )

- 平均间接效应是指所有县间接效应的平均值。

$$\text{Average indirect impact}_{\text{Texas}} = \frac{1}{254} (IDI_{\text{Dallas}} + IDI_{\text{Houston}} + \dots)$$

## 平均总体效应 (Average total impact)

- 一个县的总体效应是指所有县的失业率对本县犯罪率的边际效应的总和 (其中, 所有县包含本县)。
- 达拉斯的总体效应 (Total impact, TI)

$$\begin{aligned} \text{Total impact}_{\text{Dallas}} &= \frac{d \text{ hrate}_{\text{Dallas}}}{d \text{ unemployment}_{\text{Dallas}}} + \frac{d \text{ hrate}_{\text{Dallas}}}{d \text{ unemployment}_{\text{Houston}}} + \dots \\ &= \text{Direct impact}_{\text{Dallas}} + \text{Indirect impact}_{\text{Dallas}} \end{aligned}$$

- 平均总体效应是指平均直接效应和平均间接效应的和。

$$\text{Average total impact} = \text{Average direct impact} + \text{Average indirect impact}$$

## 使用 margins 和 predict 做政策分析

如果每个县的失业率都下降 1%，犯罪率会受到怎样的影响？

```
. margins, at (unemployment=generate(unemployment))          ///
>             at (unemployment=generate(unemployment-1))
Adjusted predictions          Number of obs          =          254
Expression   : Reduced-form mean, predict()
1._at        : unemployment      = unemployment
2._at        : unemployment      = unemployment-1
```

	Delta-method						
	Margin	Std. Err.	z	P> z	[95% Conf. Interval]		
_at							
1	8.352163	.6395964	13.06	0.000	7.098577	9.605748	
2	7.694502	.6755559	11.39	0.000	6.370437	9.018567	

- 平均来讲，犯罪率会从 8.4% 跌至 7.7%。

# 目录


1 为什么需要空间自回归模型？

2 实例：得克萨斯州犯罪率

3 Stata 之 **Sp** 系列命令的概述

# Stata 的 Sp 系列命令的概述

## 流程



## spregress: 估计横截面数据的空间自回归模型

- **spregress** 可以计算如下广义的空间自回归模型。

$$y = X\beta + \sum_k^K W_k X \gamma_k + \sum_{\rho=1}^P \lambda_{\rho} W_{\rho} y + u$$

$$u = \sum_{q=1}^Q \rho_q M_q u + \epsilon$$

- $u$  是无法观测到的误差项，允许空间相关联。
- 以上模型是广义的，比较复杂。我们为了便于理解，化繁为简地解释。

## 自变量的空间滞后 (Spatial lag of independent variables, $WX$ )

- 例子：一个县的犯罪率取决于周边县的警察分布密度。
- 程序：**spregress** 带选项 **ivarlag()**

```
spregress crime police, ivarlag(W : police) gs2sls
```

- 模型:

$$\text{crime} = \beta_0 + \beta_1 * \text{police} + \gamma * W * \text{police} + \text{errors}$$

- 空间 Durbin 模型 (spatial Durbin model):

$$y = X\beta + WX\gamma + \text{errors}$$

空间单位  $i$  的因变量  $y$  会被周边空间单位的自变量  $X$  影响，其中矩阵  $W$  定义了什么是周边空间单位。

## 因变量的空间滞后 (Spatial lag of dependent variable, $Wy$ )

- 例子：一个县的犯罪率受周边县的犯罪率的影响。一个罪犯在一个地方犯罪，有可能同样在周边地区犯罪。
- 程序: **spregress** 和选项 **dvarlag()**

```
spregress crime unemployment, dvarlag(W) gs2s1s
```

- 模型:

$$\text{crime} = \beta_0 + \beta_1 * \text{unemployment} + \lambda * W * \text{crime} + \text{errors}$$

- 空间滞后模型 (spatial lag model):

$$y = X\beta + \lambda Wy + \text{errors}$$

空间单位  $i$  的因变量  $y$  会受到其它空间单位因变量  $y$  的影响。

## 误差项的空间滞后 (Spatial lag of errors , $Mu$ )

- 例子：纽约街区的白血病发病率受周边街区化学排放污染的影响，而我们无法测量化学排放污染。
- 程序: **spregress** 和选项 **errorlag()**

```
spregress leukemia , errorlag(M) gs2s1s
```

- 模型:

$$\text{leukemia} = \beta_0 + u \quad \text{and} \quad u = \rho Mu + \epsilon$$

- 空间误差模型 (Spatial error model) :

$$y = \beta_0 + X\beta + u \quad \text{and} \quad u = \rho Mu + \epsilon$$

空间单位  $i$  的因变量  $y$  受周边地区不可观测的因素  $u$  的影响。

# 广义空间自回归模型 (General spatial autoregressive models)

- 灵感：空间的相关性是通过多个渠道发生的。因变量既取决于邻居的因变量, 又取决于邻居的自变量, 误差项取决于邻居的误差项。我们想让模型呈现出多种空间相关的渠道。
- 例子:

$$\begin{aligned} \text{crime} &= \beta_0 + \beta_1 * \text{police} + \beta_3 * \text{unemployment} \\ &\quad + \gamma * W_1 * \text{police} + \lambda * W_2 * \text{crime} + u \\ u &= \rho M u + \epsilon \end{aligned}$$

- 程序:

```
spregress crime police unemployment, gs2sls ///  
        ivarlag(W1: police) dvarlag(W2) errorlag(M)
```

## 高阶空间自回归模型

- 例子：一个县的犯罪率影响与它接壤的邻居县以及这些邻居县的邻居。我们认为空间溢出的效果对于不同的邻居是不同的，因此想在模型中体现这种差别。
- 程序：

```
/* create weighting matrix for direct neighbors */  
spmatrix create contiguity W1  
  
/* create weighting matrix for second order neighbors */  
spmatrix create contiguity W2, second  
  
/* fit higher-order spatial lag model */  
spregress crime unemployment, gs2sls dvarlag(W1) dvarlag(W2)
```

- 高阶空间滞后意味着一个模型中可以有多于一个的  $Wy$  或/和  $Mu$ 。在某些情况下是有意义的。
  - ① 更好的逼近真实的空间相关性扩散的过程，或者
  - ② 不同的空间权重矩阵定义了不同的邻居，我们想看看不同层次的邻居的空间影响。

## SAR 模型的空间溢出效应 (I)

- 广义的 SAR 模型可以简写为

$$y = X\beta + \lambda Wy + u$$

(其中  $u$  可以是空间相关联的,  $X$  可以包含  $WX$ 。)

- 把因变量  $y$  解出来为

$$y = (I - \lambda W)^{-1}(X\beta + u)$$

- 因变量的条件期望为  $E(y|X)$

$$\begin{aligned} E(y|X) &= (I - \lambda W)^{-1}X\beta \\ &= (I + \lambda W + \lambda^2 W^2 + \dots)X\beta \end{aligned}$$

## SAR 模型的空间溢出效应 (II)

如果某个空间单位  $i$  的自变量  $X_{ij}$  改变了, 那么  $y$  的预测值变化可以写为

$$\begin{aligned} E(y|X_1) - E(y|X_0) &= (I - \lambda W)^{-1} X_1 \beta - (I - \lambda W)^{-1} X_0 \beta \\ &= (I - \lambda W)^{-1} (X_1 - X_0) \beta \\ &= (I - \lambda W)^{-1} \Delta X \beta \\ &= (I + \lambda W + \lambda^2 W^2 + \dots) \Delta X \beta \\ &= \Delta X \beta + \lambda W \Delta X \beta + \lambda^2 W^2 \Delta X \beta + \dots \end{aligned}$$

- 1  $\Delta X \beta$  是空间单位  $i$  对自身的影响。
- 2  $\lambda W \Delta X \beta$  是空间单位  $i$  对它的邻居的影响。
- 3  $\lambda^2 W^2 \Delta X \beta$  是空间单位  $i$  对其邻居的邻居的影响。
- 4 以此类推其它的项。

## estat impact: 衡量空间溢出效应

- 简单来说，空间溢出效应就是指因变量  $x$  对自变量  $y$  的边际效应，即导数。
- 因为  $x$  和  $y$  是列向量， $\frac{\partial y}{\partial x}$  是一个矩阵，即雅可比矩阵 (Jacobian matrix)。

$$\frac{\partial y}{\partial x} = \begin{bmatrix} \frac{\partial y_1}{\partial x_1} & \cdots & \cdots & \cdots & \frac{\partial y_1}{\partial x_n} \\ \vdots & \ddots & & & \vdots \\ \frac{\partial y_k}{\partial x_1} & \cdots & \frac{\partial y_k}{\partial x_k} & \cdots & \frac{\partial y_k}{\partial x_n} \\ \vdots & & & \ddots & \vdots \\ \frac{\partial y_n}{\partial x_1} & \cdots & \cdots & \cdots & \frac{\partial y_n}{\partial x_n} \end{bmatrix}$$
$$= \begin{bmatrix} \frac{\partial y_1}{\partial x_1} & & & & \\ & \ddots & & & \\ & & \frac{\partial y_k}{\partial x_k} & & \\ & & & \ddots & \\ & & & & \frac{\partial y_n}{\partial x_n} \end{bmatrix} + \begin{bmatrix} 0 & \cdots & \cdots & \cdots & \frac{\partial y_1}{\partial x_n} \\ \vdots & \ddots & & & \vdots \\ \frac{\partial y_k}{\partial x_1} & \cdots & 0 & \cdots & \frac{\partial y_k}{\partial x_n} \\ \vdots & & & \ddots & \vdots \\ \frac{\partial y_n}{\partial x_1} & \cdots & \cdots & \cdots & 0 \end{bmatrix}$$

## 直接效应

直接效应就是空间单位对自身的影响。

$$\begin{bmatrix} \frac{\partial y_1}{\partial x_1} & & & & \\ & \dots & & & \\ & & \frac{\partial y_k}{\partial x_k} & & \\ & & & \dots & \\ & & & & \frac{\partial y_n}{\partial x_n} \end{bmatrix}$$

平均直接效应就是以上矩阵对角线的和的平均值。

$$\text{平均直接效应} = \frac{1}{n} \sum_{i=1}^n \frac{\partial y_i}{\partial x_i}$$

## 间接效应

间接效应是空间单位对其它周边单位的影响。

$$\begin{bmatrix} 0 & \cdots & \cdots & \cdots & \frac{\partial y_1}{\partial x_n} \\ \vdots & \ddots & & & \vdots \\ \frac{\partial y_k}{\partial x_1} & \cdots & 0 & \cdots & \frac{\partial y_k}{\partial x_n} \\ \vdots & & & \ddots & \vdots \\ \frac{\partial y_n}{\partial x_1} & \cdots & \cdots & \cdots & 0 \end{bmatrix}$$

平均间接效应就是非对角线的和的平均值。

$$\text{平均间接效应} = \frac{1}{n} \sum_{i=1}^n \sum_{j=1, j \neq i}^n \frac{\partial y_i}{\partial x_j}$$

# 总体效应

总体效应即空间单位对自身和对其它空间单位的影响。

$$\begin{bmatrix} \frac{\partial y_1}{\partial x_1} & & & & \\ & \ddots & & & \\ & & \frac{\partial y_k}{\partial x_k} & & \\ & & & \ddots & \\ & & & & \frac{\partial y_n}{\partial x_n} \end{bmatrix} + \begin{bmatrix} 0 & \cdots & \cdots & \cdots & \frac{\partial y_1}{\partial x_n} \\ \vdots & \ddots & & & \vdots \\ \frac{\partial y_k}{\partial x_1} & \cdots & 0 & \cdots & \frac{\partial y_k}{\partial x_n} \\ \vdots & & & \ddots & \vdots \\ \frac{\partial y_n}{\partial x_1} & \cdots & \cdots & \cdots & 0 \end{bmatrix}$$

平均总体效应就是所有项的平均值。

$$\begin{aligned} \text{平均总体效应} &= \frac{1}{n} \sum_{i=1}^n \sum_{j=1}^n \frac{\partial y_i}{\partial x_j} \\ &= \text{平均直接效应} + \text{平均间接效应} \end{aligned}$$

## 为什么 Stata 其它命令忽略间接效应?

$$\begin{bmatrix} \frac{\partial y_1}{\partial x_1} & & & & \\ & \ddots & & & \\ & & \frac{\partial y_k}{\partial x_k} & & \\ & & & \ddots & \\ & & & & \frac{\partial y_n}{\partial x_n} \end{bmatrix} + \begin{bmatrix} 0 & \cdots & \cdots & \cdots & \frac{\partial y_1}{\partial x_n} \\ \vdots & \ddots & & & \vdots \\ \frac{\partial y_k}{\partial x_1} & \cdots & 0 & \cdots & \frac{\partial y_k}{\partial x_n} \\ \vdots & & & \ddots & \vdots \\ \frac{\partial y_n}{\partial x_1} & \cdots & \cdots & \cdots & 0 \end{bmatrix}$$

- 对于大多数经典的计量模型来说,  $\frac{\partial y_i}{\partial x_j} = 0$ .
- 所以, 间接效应为零。
- 总体效应 = 直接效应

## spivregress: 空间 IV 回归 (spatial IV regression)

- **spivregress** 和 **spregress** 很相似，但是前者允许内生变量 (endogenous variables)。
- **spivregress** 可以计算如下的模型：

```
spivregress y1 x1 x2 (y2 y3 = z1 z2 z3),      ///  
            dvarlag(W) errorlag(M) ivarlag(W: x1)
```

$y_2$  和  $y_3$  是内生变量，而  $z_1, z_2, z_3$  是工具变量。

- 我们所讨论的可以应用于 **spregress** 的同样可以应用于 **spivregress**。
  - 1 它可以计算广义的空间自回归模型。
  - 2 允许高阶的空间滞后项。
  - 3 使用 **estat impact**, **predict**, 和 **margins** 来解释计算结果。

## spxtregress: 面板数据空间自回归模型

- **spxtregress** 用于计算面板数据的空间自回归模型。
- 它提供了固定效应 (fixed-effects) 和随机效应 (random-effects) 模型。
- 它可以计算类似于以下的模型：

```
/* fixed-effects estimation */
spxtregress y x1 x2 , fe      ///
            dvarlag(W) errorlag(M) ivarlag(W: x1)

/* random-effects estimation */
spxtregress y x1 x2 i.year , re      ///
            dvarlag(W) errorlag(M) ivarlag(W: x1)
```

- 使用 **estat impact**, **predict**, 和 **margins** 来解释计算结果。

## 更多资源

以下资源可以帮助进一步了解 Stata 的空间自回归模型。

- [webinar resource](#)
- [Stata \[SP\] 用户手册](#)

# 总结

- 空间自回归模型可以包含
  - ▶ 因变量的空间滞后（选项 **dvarlag()**）
  - ▶ 自变量的空间滞后（选项 **ivarlag()**）
  - ▶ 误差项的空间滞后（选项 **errorlag()**）
- **spregress** 和 **spivregress** 允许因变量，自变量，误差项的高阶空间滞后项。
- 内生变量（**spivregress**）
- 面板数据的空间自回归模型（**spxtregress**）
  - ▶ 固定效应模型（选项 **fe**）
  - ▶ 随机效应模型（选项 **re**）
- 分析空间溢出效应
  - ▶ 直接效应和间接效应（**estat impact**）
  - ▶ 政策分析（**margins** 和 **predict**）
- 空间权重矩阵（**spmatrix**）
- 读入 shapefile（**spshape2dta**）

谢谢！